

Problema

- Numa determinada empresa, as vendas são calculadas no final de cada dia. Elabore um algoritmo que verifique qual foi o total de vendas no mês, a média mensal e quantas vendas (dias) foram superiores a média mensal.

1-Solução

- Entrada dos valores das vendas de cada dia, uma variável para cada dia do mês;
- Calcular o total de vendas do mês;
- Calcular a média do mês;
- Para cada dia, comparar o valor da venda com a média calculada, caso a venda seja maior então adicionar 1 a variável contadora;
- Informar o total da venda do mês, a média do mês e o nr de dias que foram superiores a média.

Problema

- Numa determinada empresa, as vendas são calculadas no final de cada dia. Elabore um algoritmo que verifique qual foi o total de vendas no mês, a média mensal e quantas vendas (dias) foram superiores a média mensal.

2-Solução

- Entrada dos valores das vendas de cada dia, uma variável para todos os dias do mês;
- A cada entrada, calcular o total de vendas do mês;
- Calcular a média do mês;
- Entrada dos valores das vendas de cada dia, uma variável para todos os dias do mês;
- A cada entrada, comparar o valor da venda com a média encontrada, caso a venda seja maior então adicionar 1 a variável contadora;
- Informar o total da venda do mês, a média do mês e o nr de dias que foram superiores a média.

Problema

- Numa determinada empresa, as vendas são calculadas no final de cada dia. Elabore um algoritmo que verifique qual foi o total de vendas no mês, a média mensal e quantas vendas (dias) foram superiores a média mensal.

3-Solução


- Entrada dos valores das vendas de cada dia, uma variável para todos os dias do mês, e entrada do valor da média do mês e total do mês;
- A cada entrada, comparar o valor da venda do dia com o valor da média do mês fornecida, caso a venda seja maior então adicionar 1 a variável contadora;
- Informar o total da venda do mês, a média do mês e o nr de dias que foram superiores a média.

Problema

- Numa determinada empresa, as vendas são calculadas no final de cada dia. Elabore um algoritmo que verifique qual foi o total de vendas no mês, a média mensal e quantas vendas (dias) foram superiores a média mensal.

4-Solução

- Usar uma estrutura de dados mais elaborada.



• UEMS •

Universidade Estadual
de Mato Grosso do Sul

Universidade Estadual de Mato Grosso do Sul

Ciência da Computação

Algoritmos e Estruturas de Dados I (AED-I)

Vetor

Prof. Nilton

nilton@comp.uems.br

Sumário

- Construção de algoritmos usando vetores

Vetor

- Uma estrutura para guardar vários dados;
- Os dados devem ser do mesmo tipo;
- O vetor possui um tamanho pré-definido;
- Um dado é acessado usando um índice;
- Um índice é uma posição do vetor.

Vetor

Nota	X	X	X	X	X	X
	1	2	3	4	5	6

- No exemplo, **Nota** é um vetor com 6 posições, podendo armazenar no máximo 6 notas;
- 1, 2, 3, 4, 5, 6 são os índices do vetor. Usa-se o índice para acessar uma nota.

Declarando um vetor

```
var  
  idade:vetor[1..20] de inteiro  
  nota:vetor[1..50] de real
```

- **idade** é um vetor para guardar 20 valores inteiros
- **nota** é um vetor para guardar 50 valores reais

Guardando um valor em uma posição do vetor

var

```
idade:vetor[1..20] de inteiro  
nota:vetor[1..50] de real
```

inicio

```
//a idade fornecida pelo usuário é armazenada na 1ª posição do vetor  
leia(idade[1])
```

```
//a nota fornecida pelo usuário é armazenada na 8ª posição do vetor  
leia(nota[8])
```

```
//a idade 60 é armazenada na 5ª posição do vetor  
idade[5] <- 60
```

```
//a nota 8.4 é armazenada na 2ª posição do vetor  
nota[2] <- 8.4
```

fimalgoritmo

Mostrando um valor de uma posição do vetor

```
var
  idade:vetor[1..20] de inteiro
  nota:vetor[1..50] de real

inicio
  idade[5] <- 60
  nota[2] <- 8.4

  //mostrando o conteúdo da 5ª posição do vetor
  escreva(idade[5])

  //mostrando o conteúdo da 2ª posição do vetor
  escreva(nota[2])
fimalgoritmo
```


Acessando todas as posições de um vetor usando uma estrutura de repetição

```
var  
  idade:vetor[1..20] de inteiro  
  nota:vetor[1..50] de real  
  i:inteiro  
  
inicio  
  //obtem do usuário 20 idades e armazena no vetor  
  para i de 1 ate 20 faca  
    leia(idade[i])  
  fimpara  
  
  //obtem do usuário 50 notas e armazena no vetor  
  para i de 1 ate 50 faca  
    leia(nota[i])  
  fimpara  
fimalgoritmo
```

Exemplo 1

- Faça um algoritmo que receba 50 idades, calcule e mostre a soma dessas idades.

```
var
  idade:vetor[1..50] de inteiro
  i:inteiro
  s:inteiro
inicio
  //obtem as idades do usuário
  para i de 1 ate 50 faca
    leia(idade[i])
  fimpara

  s <- 0 //inicializa o acumulador

  //soma cada idade e armazena em s
  para i de 1 ate 50 faca
    s <- s + idade[i]
  fimpara

  //mostra a soma das idades
  escreva(s)

fimalgoritmo
```


Exemplo 2

- Faça um algoritmo que receba 50 idades e mostre as idades acima de 30 anos.

```
var
  idade:vetor[1..50] de inteiro
  i:inteiro
inicio
  //obtem as idades do usuário
  para i de 1 ate 50 faca
    leia(idade[i])
  fimpara

  //mostra as idades acima de 30 anos
  para i de 1 ate 50 faca
    se idade[i]>30 entao
      escreva(idade[i])
    fimse
  fimpara

fimalgoritmo
```

Exemplo 3

- Faça um algoritmo que receba 50 idades, calcule e mostre quantas idades estão acima de 30 anos.

```
var
  idade:vetor[1..50] de inteiro
  i:inteiro
  cont:inteiro
inicio
  //obtem as idades do usuário
  para i de 1 ate 50 faca
    leia(idade[i])
  fimpara

  cont <- 0 //inicializa o contador

  //conta as idades acima de 30 anos
  para i de 1 ate 50 faca
    se idade[i]>30 entao
      cont <- cont + 1
    fimse
  fimpara

  //mostra quantas idades acima de 30 anos
  escreva(cont)

fimalgoritmo
```


Exemplo 4

- Faça um algoritmo que receba 100 notas, calcule mostre a média dessas notas.

```
var
    nota:vetor[1..100] de real
    i:inteiro
    s, m:real
inicio
    //obtem as notas do usuário
    para i de 1 ate 100 faca
        leia(nota[i])
    fimpara

    s <- 0 //inicializa o acumulador

    //soma as notas
    para i de 1 ate 100 faca
        s <- s + nota[i]
    fimpara

    //calcula a media
    m <- s / 100

    //mostra a media
    escreva(m)
fimalgoritmo
```

Exemplo 5

- Faça um algoritmo que preencha dois vetores, X e Y, com 20 números cada. Calcule e mostre o vetor resultante Z a partir da soma de X e Y.

```
var
  x, y, z:vetor[1..20] de inteiro
  i:inteiro
inicio
  //preenche o vetor x
  para i de 1 ate 20 faca
    leia(x[i])
  fimpara
  //preenche o vetor y
  para i de 1 ate 20 faca
    leia(y[i])
  fimpara
  //gera o vetor resultante z, x+y
  para i de 1 ate 20 faca
    z[i] <- x[i] + y[i]
  fimpara
  //mostra o vetor resultante z
  para i de 1 ate 20 faca
    escreva(z[i])
  fimpara
fimalgoritmo
```


Exemplo 6

- Faça um algoritmo que preencha um vetor com 100 números e mostre o maior número.

```
var
  n:vetor[1..100] de inteiro
  i, maior:inteiro
inicio
  //preenche o vetor
  para i de 1 ate 100 faca
    leia(n[i])
  fimpara
  //encontra o maior
  maior <- n[1]
  para i de 2 ate 100 faca
    se n[i] > maior entao
      maior <- n[i]
    fimse
  fimpara
  //mostra o maior
  escreva(maior)
fimalgoritmo
```

Resolvendo 1

- Faça um algoritmo que receba 20 números e mostre os números que são ímpares.

Resolvendo 2

- Faça um algoritmo que receba 35 números e mostre os números que são divisíveis por 5.

Resolvendo 3

- Faça um algoritmo que determine quantos alunos tem media acima da média geral da turma. A quantidade de alunos é N (entre zero e 50), a média de cada aluno deve ser fornecida e a média geral da turma deve ser calculada.

Exercícios

1. Ler 20 números em um vetor A e construir um vetor B onde cada número em B é o dobro de cada número de A. Informe o novo vetor.
2. Ler 20 números em um vetor A e construir um vetor B onde cada número em B é o fatorial de cada número de A. Informe o novo vetor.
3. Ler dois vetores A e B do tipo inteiro com 30 elementos. Construir um vetor C onde cada elemento de C é a subtração do elemento correspondente de A com B, isto é, $C[1]=A[1] - B[1]$, ... Informe o novo vetor.
4. Ler dois vetores A e B do tipo inteiro com 30 elementos. Construir um vetor C da junção dos outros dois vetores. Informe o novo vetor.
5. Ler dois vetores A e B do tipo inteiro com 30 elementos. Construir um vetor C com apenas elementos repetidos entre A e B. Informe o novo vetor.
6. Ler uma palavra e escrevê-la ao contrário.

Exercícios

7. Sendo o vetor V igual a

5	10	7	6	16	3	9	2	2	1
1	2	3	4	5	6	7	8	9	10

e as variáveis X com 1 e Y com 3, escreva o valor correspondente à cada acesso ao vetor V:

a) $V[X+1]$

b) $V[Y+2]$

c) $V[X+3]$

d) $V[X*4]$

e) $V[Y*1]$

f) $V[Y*2]$

g) $V[Y*3]$

h) $V[V[X+Y]]$

i) $V[X+Y]$

j) $V[8-V[4]]$

k) $V[V[4]]$

l) $V[V[V[7]]]$

m) $V[V[1]*V[2*5]]$

n) $V[V[X+5]]$

Exercícios

8. Ler 20 números e escreva primeiro os pares e depois os ímpares.
9. Ler 20 datas de aniversário e escreva:
 - A idade em anos;
 - A idade em dias;
 - O signo zodiacal.
10. Leia 100 números e escreva a média, o maior, o menor, o percentual de números ímpares e o percentual de múltiplos de 3.
11. Construa um algoritmo que gerencie a venda de ingressos das poltronas numeradas de um teatro, que tem 100 lugares. Para cada nova reserva deve ser verificado se o assento se encontra disponível.
12. Dados 2 vetores de 20 números, utilize um terceiro vetor para intercalar os números dos dois. Armazene os números do primeiro nas posições ímpares do terceiro vetor, e os do segundo nas posições pares.